/* Copyright (c) 2006, Jan Flora <janflora@diku.dk>
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without modification,
 * are permitted provided that the following conditions are met:
 *
 *  - Redistributions of source code must retain the above copyright notice, this
 *    list of conditions and the following disclaimer.
 *  - Redistributions in binary form must reproduce the above copyright notice,
 *    this list of conditions and the following disclaimer in the documentation
 *    and/or other materials provided with the distribution.
 *  - Neither the name of the University of Copenhagen nor the names of its
 *    contributors may be used to endorse or promote products derived from this
 *    software without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND ANY
 * EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES
 * OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT
 * SHALL THE COPYRIGHT OWNER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
 * SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT
 * OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION)
 * HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR
 * TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE,
 * EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
 */

/*
  @author Jan Flora <janflora@diku.dk>
*/

#include "MacPib.h"

module FrameTxM
{
	provides
	{
		interface FrameTx[uint8_t service];
		interface Reset;
	}
	uses
	{	
		interface PhyTransmit;
		interface Debug;
	}
}
implementation
{
	#define DBG_LEVEL 1
	#include "Debug.h"
	
	#define TX_QUEUE_SIZE 10
	
	uint8_t txQueue[TX_QUEUE_SIZE];
	uint8_t nextDone = 0;
	uint8_t nextFree = 0;

	command void Reset.reset()
	{
		atomic {
			nextDone = 0;
			nextFree = 0;
		}
	}
	
	async command phy_error_t FrameTx.tx[uint8_t service](txdata_t *data)
	{
		phy_error_t myError;
		atomic {
			phyIsReceiving = FALSE;
			phyIsTransmitting = TRUE;
		
			myError = call PhyTransmit.tx(data);
			if (myError == PHY_SUCCESS) {
				// Enqueue service id.
				txQueue[nextFree] = service;
				nextFree = (nextFree+1)%TX_QUEUE_SIZE;
			}
		}
		//DBG_STRINT("Enqueued service id: ", service,1);
		//DBG_STRINT("Next slot is: ", nextFree,1);
		return myError;
	}
	
	async event void PhyTransmit.txDone(phy_error_t error)
	{
		uint8_t myService;
		
		atomic {
			phyIsTransmitting = FALSE;
			myService = txQueue[nextDone];
			nextDone = (nextDone+1)%TX_QUEUE_SIZE;
		}
		//DBG_STRINT("Done transmitting service id: ",myService,1);
		//DBG_STRINT("Next done slot is: ",nextDone,1);
		signal FrameTx.txDone[myService](error);
	}

	default async event void FrameTx.txDone[uint8_t service](phy_error_t error)
	{
		DBG_STRINT("WARNING: A txDone event was not connected. Service id was: ",service,1);
	}
}
